
Abstract. In this paper, we formalize in the Coq proof assistant an ML-
like language with imperative features, a monomorphic type system, a
reduction semantics, and the type soundness. We consider this language
as an extension of Mini-ML, and emphasize the consequences on the
definitions and the proofs due to this extension.

1 Introduction

This paper reports on the machine-checked proof of the type soundness for a
kernel of the ML language incorporating imperative features (references, assign-
ments, sequences). Type soundness is a safety property that relates typing and
evaluation : the evaluation of a well-typed program either loops, or terminates
(by computing a value). Thus it never fails because of a type error or in other
words, according to Milner,Well-typed programs do not go wrong.

Different formal methods exist to describe the semantics of a language, we
have chosen to use a reduction operational semantics (also called small-step
semantics) as it is exemplified in [16]. Such a method allows to separate the
different concepts and consequently provides more abstraction and modularity.
Furthermore it describes the computation a step at a time, and so allows to
express very fine properties about the computation. Some other works about the
mechanical verification of language properties such as [12, 5] use also a reduction
semantics.

The formal development illustrated in this paper is done within the Coq proof
assistant [6] and consequently can be seen as another formal piece that supple-
ments the certification laid out in [5, 4] even if we consider here a monomorphic
type system.

As far as we are aware, among the different computer-verified proofs (e.g.
[5, 4, 3, 12, 9, 15, 11, 13]), no publication mentions the computer-verified for-
malization and proof of type soundness for a fragment of ML mixing impera-
tive and functional primitives. In [14], VanInwegen presents a formalization of
Core Standard ML within HOL and by the way incorporates references. She has
proved the type preservation property (it means that when the evaluation of a
well-typed program terminates, it computes a value of the same type than the
type of the program) for a good portion of the language. However this work is
done with a big step semantics (based on an evaluation semantic relation).

Beyond this important and useful purpose, our work is the starting-point
of the development of a formal framework (based on the Calculus of Inductive
Constructions and Coq) that allows to define type systems à la ML and to
reason about them. Such a project requires at the same time the construction
of formal components, a methodology and tools for composing and re-using
formal pieces. At the present time, in order to reach this long term objective, we
compare the formal development of the type soundness for Mini-ML, a language
without imperative aspects and the formal development of the same property for
Reference-ML, the same language with imperative features. Along this paper,
we’ll emphasize what have to be added in the proof or re-defined.

We assume here familiarity with the Calculus of Inductive Constructions.
We use version 6.3 of the Coq proof assistant. In order to make this paper
more readable, we adopt a pseudo-Coq syntax which differs slightly from the
usual Coq syntax. Our paper provides the definitions of most concepts, the key
lemmas but almost no proofs. The complete development is accessible on the
Internet via http://www.iie.cnam.fr/~boite/monomorphic soundness.tgz.

The next section introduces a generic structure used to represent type envi-
ronment and store. Section 3 describes the language. The semantics is exposed
and formalized in the fourth section. Then, we present the typing in section 5.
In the last section, we deal with the type soundness.

2 A Generic Table

Static and dynamic semantics use structures to store information, for instance
type environments that map free identifiers to types, or memories that map
locations to values. So we present in this section the formalization of a generic
structure table, which has two implicit parameters A and B, where A is the
type of the keys, and B the type of the stored values. We define it as an inductive
type with only one constructor intro table : its parameters are the domain of
the table represented by a list of elements of type A, and a function from A to
B. The type A is required to verify the decidability of the equality (predicate
eq A dec).

Inductive table : Set :=
intro_table : (list A) → (A → B) → table.

An object of type (table A B) is in fact a function whose domain is explicitly
given, which allows us to use it as a partial function, or an association table.
We define the operations of application, domain computing, adding of a new
information in the table. We see a lot of advantages compared to a list of pairs
: adding a new association doesn’t create any redundancy, the computation of
the domain and the application are got for “free”, as we can see below.

Inductive apply_table : (table A B)→ A → B → Prop :=
intro_apply_table : ∀l:A,∀dom:(list A), ∀f:A→B,
l∈dom→(apply_table (intro_table dom f) l (f l)).

The predicate (apply table t a b) stands for two properties : a ∈ dom(t)
and t(a) = b, so the partiality of the function isn’t a problem. The operation of
adding is less simple, as we avoid redundancy.

Definition add_table [a:A; b:B; t:(table A B)] : table :=
Cases t of (intro_table dom f) ⇒ Cases a∈dom of
_ ⇒(intro_table dom
(λa’:A.(Cases (eq_A_dec a a’) of _ ⇒ b | _ ⇒ (f a’)))

| _⇒(intro_table (Cons a dom)
(λa’:A.(Cases (eq_A_dec a a’) of _ ⇒ b | _ ⇒ (f a’)))

The concrete definition of a table is never used in the formalization. We ma-
nipulate a table as an abstract data via several lemmas. The two most important
of them follow, and are used to define a equivalence relation between tables.

Lemma swap_table : ∀A,B:Set,∀a,a’,e:A,∀b,b’,r:B,∀t:table,
~a=a’ → (apply_table (add_table a’ b’ (add_table a b t)) e r)→
(apply_table (add_table a b (add_table a’ b’ t)) e r).

Lemma add_add_table : ∀A,B:Set,∀t:table,∀a,e:A,∀b,b’,r:B,
(apply_table (add_table a b t) e r) →
(apply_table (add_table a b (add_table a b’ t)) e r).

The total Coq script for this structure with its lemmas overtakes 1000 lines.

3 The Language Reference-ML

The first step to study a language is to formally define its syntax. However, it is
not independent from the semantics. Actually, the semantics imposes to define
two notions in addition to the language: memory locations and values. We first
detail the syntax of the language. Then, in the second subsection, we discuss
semantic extensions, and lastly we give the Coq translation.

3.1 Expressions

We consider the following grammar of the expressions, in which we emphasize
in a bold font the constructions of Reference-ML added to Mini-ML :

e ::= c constant | e ; e sequence
| x identifier | while e do e loop
| e e application | ref e reference creation
| fun x e abstraction | !e look-up
| let x=e in e local definition | e:=e assignment
| if e then e else e conditional

c ::= n integer
| b boolean
| unit side-effect result

So Reference-ML is an extended λ-calculus which incorporates references and
constructions to manipulate them. A reference is a memory cell containing a
value. Its contains may be modified all along the program by using assignments.

3.2 Semantic Extensions

We’ve made the choice to work with a reduction semantics, also called Small-Step
semantics. It focuses on each elementary step calculus, by rewriting expressions.
With references, we have to take care of the memory state, which can change dur-
ing the reduction process. For instance, the evaluation of an assignment e1 := e2

needs to evaluate e1 to compute the location that is the address of the cell, whose
contents in the memory has to be modified, and needs to evaluate e2 to know
which value to put in the memory. So, we have to explicitly manipulate loca-
tions in the reduction steps. The notion of location doesn’t belong to the user’s
abstract syntax, but as we need it in the reduction process, we’ve introduced it
in the Coq definition of the language.

Among the expressions, we have to distinguish the values. This notion is syn-
tactic : a value is either a constant, either an abstraction (because the reduction
strategy we use is a head reduction), or a location.

3.3 Coq Formalization

In order to name the identifiers and the locations, we define two abstract types,
respectively identifier and location. We assume the decidability of equality
on both types, and we assume that one can always find a fresh location wrt a
memory, that is to say one can exhibit a location l so that l denotes an address
that does not appear in the memory.

The language specification in Coq is obviously an inductive type :

Inductive Constant : Set :=
Const_int : nat → Constant

| Const_bool: bool → Constant
| unit : Constant.

Inductive expr : Set :=
Const : Constant → expr

| Var : identifier → expr
| Loc : location → expr
| Fun : identifier → expr → expr
| Apply : expr → expr → expr
| Let_in: identifier → expr → expr → expr
| If : expr → expr → expr → expr
| While : expr → expr → expr
| Ref : expr → expr
| Deref : expr → expr
| Sequ : expr → expr → expr
| Assign: expr → expr → expr.



Type Soundness of a Simply Typed ML-Like Language with References 73

We define an object of type value as a pair composed of an expression e and
a proof that e is a value.

Inductive is_value : expr → Prop :=
Cst_val : ∀c:Constant,(is_value (Const c))

| Loc_val : ∀l:location,(is_value (Loc l))
| Fun_val : ∀i:identifier, ∀e:expr,(is_value (Fun i e)).

Inductive value : Set :=
value_intro : ∀e:expr,(is_value e)→value.

3.4 From Mini-ML to Reference ML

The extension of the language consists simply in adding new constructors in
the inductive types expr and Constant : Loc, Sequ, While, Ref, Deref,
Assign, and unit. So going from Mini-ML to Reference-ML produces super-
types (in the sense of object-oriented programming) : any Mini-ML expression
is also a Reference-ML expression.

The predicate is value is extended with the Loc val clause, but the defini-
tion of the type value is unchanged.

4 Reduction Semantics

4.1 Elementary Reduction and Reduction Strategy

As a reduction semantics takes care of elementary steps of the computation,
it’s straightforward to explain the memory evolution during the reduction. A
memory state is a finite mapping between locations and values. We require three
operations on a memory m :

• dom(m) which computes the set of locations in the memory.
• m(l) which gives the value mapped to l in m.
• m⊕ (l, v) which maps l to v, and maps l′ to m(l′) if l′ ∈ dom(m) and l′ 6= l.

A configuration is an (expression/memory state)-pair, and the reduction relation
is a transition relation between configurations. A reduction of the expression
a1 in the memory state m1, into a2 in the memory state m2 will be noted
a1/m1 −→ a2/m2.

To define the reduction strategy, we’ve adopted the technique which con-
sists in defining an elementary reduction, also called ε-reduction, and the no-
tion of reduction context, as proposed in [16]. We define the ε-reduction noted
a1/m1 −→ε a2/m2 in Fig.1. In all the rules, v denotes a value. The two first
rules corresponds to β-reductions, and a[v/x] denotes the substitution of x by v
in a.

(fun x a) v / m −→ε a[v/x] / m
let x = v in a / m −→ε a[v/x] / m

if true then e1 else e2 / m −→ε e1 / m
if false then e1 else e2 / m −→ε e2 / m

while e1 do e2 / m −→ε if e1 then (e2 ; while e1 do e2) else unit / m
ref v / m −→ε l / m⊕ (l, v) if l /∈ dom(m)
l := v / m −→ε unit / m⊕ (l, v) if l ∈ dom(m)

!l /m −→ε m(l) / m if l ∈ dom(m)
v; e/ m −→ε e/ m

Fig. 1. ε-reduction rules

A reduction context E can be seen as an expression with a unique hole, and
is defined as follows :

E ::= • | E e | v E | let x = E in e | if E then e else e
| E; e | ref E | E := e | !E | l := E

where • is a hole, e an expression, v a value, and l a location. E[e] denotes E
where the hole has been filled with e.

The relation −→ is defined as the smallest relation containing −→ε and
satisfying the rule Context given below that allows to reduce in a context :

a1/m1−→ a2/m2

E[a1]/m1−→E[a2]/m2
(Context)

4.2 Substitution

The β-reductions in Fig.1 make appear substitutions of an identifier by an ex-
pression in another one. The substitution operation must ensure not to substi-
tute bound variables, neither to capture variables. Modelling the operation of
substitution by a function would require to rename some bound variables to
prevent captures, so we have represented this operation in Coq with a predicate
(subst expr e x e’ e’’) which means e′′ = e[e′\x]. This predicate is defined
by case analysis on e, and it uses the predicate free ident to know if an identi-
fier is free or not in a given expression. This last one is also defined by induction
on the expression. A fragment of the code is given below :

Inductive free_ident : identifier → expr → Prop :=
free_var : ∀x:identifier,(free_ident x (Var x))
|free_apply1 : ∀x:identifier,∀e1,e2:expr,(free_ident x e1)→

(free_ident x (Apply e1 e2))
|free_apply2 : ∀x:identifier,∀e1,e2:expr,(free_ident x e2)→

(free_ident x (Apply e1 e2))
...

Inductive subst_expr : expr→identifier→expr→expr→Prop :=
subst_const: ∀x:identifier,∀e’:expr,∀c:Constant

(subst_expr (Const c) x e’ (Const c))
|subst_var : ∀x:identifier,∀e’:expr

(subst_expr (Var x) x e’ e’)
|subst_nvar : ∀x,y:identifier,∀e’:expr, ¬x=y →

(subst_expr (Var y) x e’ (Var y))
| subst_fun : ∀x:identifier,∀e,e’:expr,

(subst_expr (Fun x e) x e’ (Fun x e))
...

4.3 Formalization of State Memory

A memory is an association table from keys of type location to objects of type
value. So, to represent a memory, we use the generic structure table defined in
section 2.

Definition memory := (table location value).

4.4 The Reduction Relation in Coq

We follow the informal presentation given in section 4.1. So, we begin to define
−→ε as the inductive predicate red epsilon.

Definition configuration := expr*memory.

Inductive red_epsilon : configuration → configuration → Prop :=
beta1 : ∀x:identifier,∀e,er,v:expr,∀m:memory,

(is_value v)→ (subst_expr e x v er)→
(red_epsilon ((Apply (Fun x e) v),m) (er,m))

| beta2 : ∀x:identifier,∀e,er,v:expr,∀m:memory,
(is_value v)→ (subst_expr e x v er)→
(red_epsilon ((Let_in x v e),m) (er,m))

| red_ref:∀m:memory,∀v:expr,∀l:location,∀V:(is_value v))
¬(in_dom l m)→
(red_epsilon ((Ref v),m)

((Loc l),(add_table l (value_intro v V) m)))
...

When we want to reduce e in the context E, we have to reduce the expression
E in which the hole has been replaced by e. It means we need a mechanism to fill
a hole. It’s interesting and practical to represent a context by a function expr
→ expr. In that case, we benefit from the functionality of Coq, and filling a
hole is simply applying a function. However all the functions of type expr →
expr are not valid contexts. We precise the set of valid contexts with the help of
the inductive predicate is context. This technique was previously introduced
by one of the authors in [5].

Definition context := expr → expr.

Inductive is_context : context → Prop :=
hole : (is_context (λx:expr. x))

| app_left : (e:expr)(E:context) (is_context E) →
(is_context (λx:expr. (Apply (E x) e)))

| app_right : (v:expr)(E:context) (is_context E) → (is_value v)→
(is_context (λx:expr. (Apply v (E x))))

...

Now, the reduction relation −→ in Coq is simply :

Inductive red : configuration → configuration → Prop :=
epsilon : ∀c1,c2:configuration,

(red_epsilon c1 c2)→(red c1 c2)
| cont : ∀c1,c2:configuration,∀E:context, (red c1 c2) →

(is_context E)→
(red (app_ctx E c1) (app_ctx E c2)).

where app ctx is a function which applies a context E on a configuration c, that
is computes E[c].

Definition app_ctx :=
λE:context.λc:configuration.
Cases c of (e,m) ⇒ ((E e),m) end.

4.5 From Mini-ML to Reference-ML

The relation red rewrites a configuration into another configuration. In Mini-ML,
a configuration is an expression, whereas in Reference-ML, a configuration is a
(expression/memory state)-pair. By re-defining configuration and the function
app ctx, the relation red is binary in both cases, and its definition has exactly
the same formulation in both languages.

The relation red epsilon in Reference-ML can be considered as a conser-
vative extension of the Mini-ML one : five rules are added for the imperative
features, and the old rules are the same (except the supplementary parameter,
a memory state, that doesn’t change in these rules).

The predicates is context, free ident, and subst expr are simply ex-
tended with the new cases.

5 Typing

5.1 The Types

The type algebra of our language is defined as follows :

τ ::= basic types | α | τ → τ | τ Ref

The basic types are Int, Bool, and Unit. The symbol α denotes a type variable.
The type τ → τ denotes the functional type, and τ Ref is the type of a refer-
ence that contains a value of type τ . Here again, the constructors required for
Reference-ML are printed in the bold font.

We assume a set stamp for the type variables, and the translation within
Coq introduces the simple following inductive types :

Inductive Basic : Set := Int : Basic | Bool : Basic | Unit : Basic.

Inductive type : Set :=
Const_t: Basic → type

| Var_t: stamp → type
| Arrow: type → type → type
| Ref_t: type → type.

5.2 The Typing Rules

We define the typing rules, with a set of inference rules, given in a natural
deduction style [7], between judgments Γ ` e : τ meaning that in the type
environment Γ , the expression e has the type τ . The function type of constant
relates a constant with its type. The typing rules in Fig.2 define the well-typed
expressions of the language. The rules in the right side of the figure are specific
to Reference-ML. We also have to give the typing rule (Loc) for locations: they
can appear during the reduction process and thus, it is necessary to be able to
type expressions that contain locations.

Γ`c : type of constant(c) (Cst)
Γ`e : τ

Γ`ref e : τ Ref (Ref)

Γ (x)=τ
Γ`x : τ (Var)

Γ`e : τ Ref
Γ`!e : τ (Deref)

Γ⊕(x:τ)`e : τ ′

Γ`fun x e : τ→τ ′ (Fun)
Γ`e1 : τ ′ Γ`e2 : τ

Γ`e1 ; e2 : τ (Seq)

Γ`e : τ→τ ′ Γ`e′ : τ
Γ`e e′ : τ ′ (App)

Γ`e1 : τ Ref Γ`e2 : τ
Γ`e1 := e2 : Unit (Assign)

Γ`e : τ Γ⊕(x:τ)`e′ : τ ′

Γ`let x=e in e′ : τ ′ (Let)
Γ`e1 : Bool Γ`e2 : τ
Γ`while e1 do e2 : Unit (While)

Γ`e1 : Bool Γ`e2 : τ Γ`e3 : τ
Γ`if e1 then e2 else e3 : τ (If)

Γ (l)=τ
Γ`l : τ Ref (Loc)

Fig. 2. Typing rules

In this context, type environments contain not only type information about
free identifiers, but also type information about locations. A type environment
may be considered as a pair (Γid,Γloc) :

Γid ::= ∅ | Γid ⊕ (x : τ) Γloc ::= ∅ | Γloc ⊕ (l : τ)

made up of a part Γid relative to identifiers, and a part Γloc relative to
locations. Each part of the pair is an association table, and is specified in Coq
by the generic structure table with the necessary parameters. This decomposition
in the implementation makes easier the treatment.

So Γid, Γloc, and Γ are respectively specified by

Definition typ_env_ident := (table identifier type).
Definition typ_env_loc := (table location type).
Definition typ_env := typ_env_ident*typ_env_loc.

We require two operations on type environments : the updating ⊕ and the
application, using the corresponding operations on tables :

(Γid, Γloc)⊕ (k : τ) = (Γid ⊕ (k : τ), Γloc) if k is an identifier
= (Γid, Γloc ⊕ (k : τ)) if k is a location

(Γid, Γloc)(k) = Γid(k) if k is an identifier
= Γloc(k) if k is a location

To define the typing rules in Coq, we use an inductive predicate that contains
a constructor per typing rule :

Inductive type_of : typ_env → expr → type → Prop :=
type_of_const : ∀ Γ:typ_env,∀c:Constant

(type_of Γ (Const c) (type_of_constant c)))
|type_of_var: ∀ Γ:typ_env_ident,∀x:identifier,∀ τ:type

Γ(x)=τ → (type_of Γ (Var x) τ)
...

5.3 Configuration Types

Nothing ensures for a given location that the type of the value associated in the
memory, and the type of this location in the type environment are the same.
So, we define the notion of well-typed memory in an environment : the domain
of the memory m and the domain of the locations part of the type environment
Γ must be the same, and for each location l of this domain, if the type of l is
τ Ref in the environment, then the value of l in the memory has the type τ in
the environment. The Coq definition memory respects env follows this informal
specification. To denote this relation informally, we write Γ ` m.

Definition memory_respects_env [Γ:typ_env;m:memory] : Prop :=
Cases Γ of (Γid,Γloc) ⇒

(∀l:location,∀τ:type,v:expr,(apply_table m l v) →
(apply_table Γloc l τ) → (type_of Γ (value2expr v) τ))
∧ dom(m)=dom(Γloc)

A configuration a/m is well-typed in an environment Γ , if a is well-typed
in Γ and if m respects Γ . We define the typing of a configuration in Γ as the
conjunction of the typing of the expression and the well-typing of the memory,
both in Γ .

Definition type_of_config [Γ:typ_env;c:configuration;τ:type] :=
Cases c of (e,m) ⇒
(type_of Γ e τ) ∧ (memory_respects_env Γ m)

5.4 From Mini-ML to Reference ML

The type algebra is extended with the type for references. Again we define a
supertype. For the constructors common to Mini-ML and Reference-ML, the
typing rules remains identical up to the re-definition of the operations ⊕ and
application on environments. The typing relation we want for Reference-ML
has to type configurations - not only expressions. It means the memory has to
respect the type environment, so we add this condition to the type relation as a
conjunction.

6 Type Soundness

Among the syntactically correct programs, we can distinguish three groups, those
whose evaluation :

• terminates on a value
• indefinitely loops
• blocks (i.e. can’t be reduced anymore but is not a value)

The role of the typing is to limit the set of syntactically correct programs. We
describe in this section the proof of the safety (or the soundness) of our type
system. It means that the evaluation of a well-typed program, if it terminates,
is a value. In other words, no type errors can appear during the computation,
and the typing eliminates blocking-programs.

We follow the structure of the paper-and-pencil proof given in [8], adapted
to the monomorphic case.

6.1 The less typable Relation

We define an ordering between configurations as follows :
a1/m1 is less typable than a2/m2, written a1/m1 v a2/m2, if for all environment
Γ and all type τ , there exists an extension Γ ′ of Γ so that

(Γ ` a1 : τ ∧ Γ ` m1) =⇒ (Γ ′ ` a2 : τ ∧ Γ ′ ` m2)

For instance, let x = (ref 1) in !x / ∅ v !l / (l : 1).

Here is the Coq specification of v :

Definition less_typable [c1,c2:configuration]: Prop :=
∀Γ:typ_env,∀τ:type,
(type_of_config Γ c1 τ) →
(∃Γ’:typ_env, (extend Γ’ Γ) ∧ (type_of_config Γ’ c2 τ)).

The extension of an environment consists in adding type information about
new locations. So, Γ ′ extends Γ , if Γ ′=Γ ⊕ (l1 : τ1) ⊕ . . . ⊕ (ln : τn), with
li /∈ dom(Γ ). We denote it Γ ′ À Γ . For the Coq specification, we require the
inclusion of the locations parts of the environments, and the equality on the
identifier parts.

Definition extend [Γ’,Γ:typ_env] : Prop :=
Cases Γ’ Γ of
(Γid’,Γloc’) (Γid,Γloc) ⇒ Γid’=Γid →

∀l:location,∀τ:type,(l ∈ Γloc)→
(apply_table Γloc’ l τ)→(apply_table Γloc l τ)

6.2 Subject Reduction

This theorem means the reduction preserves the type, and it ensures the com-
putation won’t create any type error.

Theorem 1 (Subject Reduction).
If a1/m1 −→ a2/m2, then a1/m1 v a2/m2

To prove this theorem, we use the type preservation by ε-reduction, and the
growing of v by context application. Then the proof of Subject Reduction holds
in three Coq lines :

Theorem Subject_reduction : ∀c1,c2:configuration,
(red c1 c2) → (less_typable c1 c2).

Induction 1;Intros.
Apply type_preservation_by_red_epsilon; Assumption.
Apply less_typable_grows; Assumption.

Proposition 1 (Type preservation by epsilon-reduction).
if a1/m1 −→ ε a2/m2, then a1/m1 v a2/m2

Lemma type_preservation_by_red_epsilon : ∀c1,c2:configuration,
(red_epsilon c1 c2) → (less_typable c1 c2).

The proof is done by case analysis on the ε-reduction rule. In each case, we
give the extended environment which allows to type the configuration c2. In the
β-reduction cases, we use the following lemma :

Lemma 1 (Substitution lemma).
if Γ ` e : τ and Γ ⊕ (i, τ) ` e1 : τ1 then Γ ` e1[i\e] : τ1

translated into Coq by :

Lemma substitution_lemma:
∀e1,e2,e:expr,∀i:identifier,∀Γ:typ_env,∀τ,τ1:type,
(subst_expr e1 i e e2)→
(type_of Γ ⊕ (i : τ) e1 τ1) →

(type_of Γ e τ) →
(type_of Γ e2 τ1).

This lemma is proved by induction on the expression e1. The proof is easy, but
requires to show that if an expression is well-typed in an environment then it
is well-typed in an environment equivalent to the first one and furthermore the
type is the same.

Proposition 2 (Growing of v).
if a1/m1 v a2/m2, then E[a1]/m1 v E[a2]/m2

Lemma less_typable_grows :∀E:context,∀c1,c2:configuration,
(is_context E) → (less_typable c1 c2) →

(less_typable (app_ctx E c1) (app_ctx E c2)).

The proof is a structural induction on the predicate is context defining E.
For each case, we give the extended environment Γ ′ of Γ , given by the hypothesis
(less typable c1 c2), to prove Γ ′ ` E[e2] : τ with the condition Γ ` E[e1] : τ .

6.3 Normal Form Theorem

The normal form theorem establishes that an expression in normal form, well-
typed in an environment which types no identifier, is a value. We allow to have
locations in the typing environment, because a location is a value, and to type
a location we need the type information in the typing environment.

If an expression is not in normal form, it means it can be reduced. That is to
say, there exists a configuration that the relation red can reach. It is specified
in Coq with :

Definition is_reducible [c:configuration] :=
(∃c’:configuration. (red c c’)).

Definition config_is_value [c:configuration] :=
Cases c of (e,m) ⇒ (is_value e) end.

Theorem nf_typed_are_value:∀Γloc:typ_env_loc,∀τ:type,∀c:configuration,
(type_of_config (∅,Γloc) c τ) → ¬(is_reducible c) →

(config_is_value c).

The normal form theorem is a consequence of the progression lemma :

Lemma 2 (Progression).
If an environment types only locations, if a/m is well-typed in this environment,
and if a is not a value, then a can be reduced.

Lemma progression : ∀Γloc:typ_env_loc,∀c:configuration,∀τ:type,
(type_of_config (∅,Γloc) c τ) → ¬(config_is_value c) →
(is_reducible c).

We prove it by induction on a where c = a/m. When a is not a value, we
explicitly give the expression in which a can reduce. In the case of a β-redex,
the expression in which it reduces makes appear an expression resulting from
a substitution. We have to prove that this expression really exists (because the
substitution is not a total function). It is expressed by the following lemma,
established by induction on the expression e.

Lemma substitute_succeeds : ∀e,e1:expr,i:identifier,∀τ:type,
∀Γloc:typ_env_loc, (type_of (∅,Γloc) e1 τ) →
(is_value e1) → (∃ e’:expr. (subst_expr e i e1 e’)).

6.4 Strong Type Soundness

The theorem Subject Reduction ensures types are preserved across a reduction
step. The theorem of the normal form ensures a well-typed program in the initial
environment, which can’t reduce anymore, is a value. As a consequence of this
two previous theorems, we obtain the type soundness : the evaluation of a term
of type τ doesn’t block, but furthermore if it terminates, we obtain a value of
type τ . It’s the strong version of the type soundness.

Formally, we have to define the reflexive and transitive closure of the reduc-
tion relation red star :

Inductive red_star : configuration → configuration → Prop:=
red_0 : ∀sn:configuration,(red_star sn sn)

| red_n : ∀sn,s0,s’:configuration, (red s0 s’)→(red_star s’ sn)→
(red_star s0 sn).

The type soundness is specified as follows :

Theorem type_safe : ∀a:expr,∀τ:type,∀c’:configuration,
(red_star (a,∅) c’) → (type_of_config (∅,∅) (a,∅) τ) →
¬(is_reductible c’) →
(config_is_value c’) ∧ (∃env:typ_env (type_of_config env s’ t))

This theorem is proved by induction on the length of the reduction. In the case
corresponding to red 0, we apply the normal form theorem. In the induction
case red n, the property follows from the induction hypothesis and the Subject
Reduction theorem.

6.5 From Mini-ML to Reference ML

In Mini-ML, a1 v a2 means if a1 has the type τ in an environment Γ , then a2 has
the type τ in Γ . The relation v of Reference-ML is an extension of that of Mini-
ML, if we re-define type of by type of config and if we accept to extend the
type environment to type the second configuration. It is a conservative extension,
because the extension of an environment concerns the location part, and there
isn’t location part in Mini-ML.

The proofs of Type preservation by ε-reduction, substitution lemma, grow-
ing of v, progression and substitute succeeds, nf typed are values are in-
ductive proofs. All the proof cases we had in Mini-ML are quite the same in
Reference-ML, and of course, new cases are treated.

The proof of Subject Reduction, as it uses the previous propositions - is ex-
actly the same in both languages. Our final theorem, type-safe, concludes
(config is value c’). It is the same theorem as in Mini-ML if we re-define
config is value, and the proof (by induction) has the same cases as Mini-ML
plus its own cases.

7 Conclusion

This work has two interests. The first is that, as far as we know, it is the first
machine-ckecking of the type-soundness of an ML-like language with references
in a reduction semantics. The second is that it focuses on the impact on the
definitions and the proofs when the language is extended. We can find in many
papers where a language is extended “the proof is similar to the previous case ...”.
We wanted to quantify this fact. Our experience is that the machine-checking
verifies this, if we take care to extend and re-define the good notions. In our
study, our Coq script rises from 2100 to 3700 lines (both including 1000 lines
for the package on the generic table). The paper tries to trace the different
supertypes, extensions and redefinitions that the new features introduce.

In a future work, we’ll extend Reference-ML in another way, with polymor-
phism, or object for example. Paper-pencil proofs exist for many years. All have
proved that some notions are exactly the same, others are extended, others are
re-defined. Indeed, our final aim is to define a formal method to specify the ex-
tension Li+1 of a language Li, and to use the proofs of the properties of Li to
show those of Li+1. This method wouldn’t be valid in a non-conservative exten-
sion. Several works [1], [2], [10], deal with proof reuse in the context of inductive
types, it would be very interesting to try to merge them with our concerns and
develop ad hoc tools.

Sémantique Naturelle. Thèse de Doctorat, Ecole Nationale des Ponts et Chaussées,
1995.

[14] Myra VanInwegen. Towards Type Preservation in Core SML. Technical report,
Cambridge University, 1997.

[15] T. Nipkow W. Naraschewski. Type Inference Verified: Algorithm W in Is-
abelle/HOL. Journal of Automated Reasoning, Vol 23, nos 3-4, 299-318, 1999.

[16] Wright and Felleisen. A Syntactic Approach To Type Soundness. Information
and computation, 115(1):38–94, 1994.


